VBA Code Sample #56 ( Getting a Contact Address from the Address Book

Created 7-26-1998

Helen Feddema (hfeddema@hvc.rr.com)

Instructions

This code sample shows how to pop up the Address Book from a Word template's AutoNew procedure; information from several fields of the selected address are stored in the document's custom doc properties, and displayed in fields on the document.

To test the code sample, follow the steps below:

1. Double-click on the Get Address from Contact.dot file to open an instance of the template.

2. The Address Book dialog should pop up (from the template's AutoNew procedure).

3. Select Contacts from the list of available address books.

4. Select an address for the letter, and click the OK button.

5. The name, address and phone fields should appear on the letter (of course, they may not all contain data).

Code

Public Sub AutoNew()

'Written by Helen Feddema 7-12-98

'Last modified 7-26-98

On Error GoTo AutoNewError

   Dim strToAddress As String

   Dim strToName As String

   Dim strToCompany As String

   Dim strToPhone As String

   Dim strToFax As String

   Dim prps As Object

   'Pop up address book to allow user to select a contact

   strToName = Application.GetAddress(Name:="", _

      AddressProperties:="FullName", UseAutoText:=True)

   Debug.Print "Name: " & strToName

   strToCompany = Application.GetAddress(Name:=strToName, _

      AddressProperties:="CompanyName", UseAutoText:=True, _

      displayselectdialog:=0)

   Debug.Print "Company: " & strToCompany

   strToPhone = Application.GetAddress(Name:=strToName, _

      AddressProperties:="Phone", UseAutoText:=True, _

      displayselectdialog:=0)

   Debug.Print "Phone: " & strToPhone

   strToFax = Application.GetAddress(Name:=strToName, _

      AddressProperties:="Fax", UseAutoText:=True, _

      displayselectdialog:=0)

   Debug.Print "Fax: " & strToFax

   strToAddress = Application.GetAddress(Name:=strToName, _

      AddressProperties:="Address", UseAutoText:=True, _

      displayselectdialog:=0)

   Debug.Print "Address: " & strToAddress

   'Assign address entry fields to doc properties

   Set prps = ActiveDocument.CustomDocumentProperties

   prps.Item("ToName").Value = strToName

   prps.Item("ToCompany").Value = strToCompany

   prps.Item("ToPhone").Value = strToPhone

   prps.Item("ToFax").Value = strToFax

   prps.Item("ToAddress").Value = strToAddress

   'Assign today's date to property

   prps.Item("DateCreated").Value = Format(Now(), "Short Date")

   'Update fields in document

   Selection.WholeStory

   Selection.Fields.Update

   Selection.HomeKey Unit:=wdStory

AutoNewExit:

   Exit Sub

AutoNewError:

   MsgBox "Error No.: " & Error.Number & "; Description: " & Error.Description

   Resume AutoNewExit

End Sub

Notes

· Although the Contacts address book is shown, you can actually select any available address book from the Address Book dialog's drop-down list.

· There are Debug.Print statements in the code, so you can check the contents of the fields in case of problems.

· The calls to GetAddressBook use AutoText entries.  Each entry uses one or more MAPI fields.  The Help topic for GetAddress is inaccurate.  It gives examples using double quotes, vb constants, and ampersands.  This doesn't work – they will appear literally in your output variables.  Instead, just put the field(s) you want in the AutoText entry (enclosed in angle brackets), together with any literals (commas, spaces, etc.) you want to appear in the output variable.

· For the syntax of the MAPI fields, see the AddAddress Help topic.
Contents of Zip File

	File Name
	File Type
	Destination Folder

	Get Address from Contact.dot
	Word 97 template
	\Templates\Outlook (or wherever you want)

	VBA Code Sample #56 – Getting a Contact Address from the Address Book
	Word 97 document
	\My Documents\Outlook (or wherever you want)

